Topic 2: Basic Modelling
(Version of 13th October 2017)

Pierre Flener and Jean-Noël Monette

ASTRA Research Group
on Combinatorial Optimisation
Uppsala University
Sweden

Course 1DL448:
Modelling for Combinatorial Optimisation

UPPSALA UNIVERSITET

Outline

1. The MiniZinc Language
2. Modelling
3. Reification
4. Set Variables &Constraints

UPPSALA UNIVERSITET

Outline

1. The MiniZinc Language
2. Modelling
3. Reification
4. Set Variables &Constraints

MiniZinc Model

A MiniZinc model may comprise the following items:

- Parameter declarations
- Variable declarations
- Predicate and function definitions
- Constraints
- Objective
- Output command

Types for Parameters

MiniZinc is strongly typed. The parameter types are:

- `int`: integer
- `bool`: Boolean
- `enum`: enumeration
- `float`: floating-point number
- `string`: string of characters
- `set of` τ: set of elements of type τ, which is `int`, `bool`, `enum`, `float`, or `string`
- `array[`ρ`] of` τ: possibly multidimensional array of elements of type τ, which is anything except an array; each range in ρ is an integer interval of the form $\alpha..\beta$

Example

The parameter declaration `int: n` declares an integer parameter of identifier `n`. One can also write `par int: n` in order to emphasise that `n` is a parameter.

Types for Variables

Decision variables are implicitly existentially quantified: the aim is to find satisfying (optimal) values in their domains.
The variable types for (decision) variables are:

- `int`: integer
- `bool`: Boolean
- `enum`: enumeration
- `float`: floating-point number (not used in this course)
- `set of enum` and `set of int`: set

A possibly multidimensional `array` can be declared to have variables of any variable type, but it is itself not a variable.

Example

The variable declaration `var int: n` declares a decision variable of domain `int` and identifier `n`.

Tight variable domains may accelerate the solving: see the next slides.

Literals

The following literals can be used:

- Boolean: `true` and `false`
- Integers: in decimal, hexadecimal, or octal format
- Sets: between curly braces, for example `{1,3,5}`, or as intervals, for example `10..30`
- 1D arrays: between square brackets, say `[6,3,1,7]`
- 2D arrays: A vertical bar `|` is used before the first row, between rows, and after the last row; for example `[|11,12,13,14|21,22,23,24|31,32,33,34|]`
- For higher-dimensional arrays, see slide 11

Careful: The indices of arrays start from `1` by default.

Declarations of Parameters and Variables

```
1 int: n = 4;
2 int: p;
3 p = 10;
4 set of int: Primes = {2,3,5,7,11,13};
5 var int: x;
6 var 0..23: hour = x + n;
7 var set of Primes: Taken;
```

- A parameter must be instantiated, once, to a literal; its declaration can be separated from its instantiation in the model (`p`), a datafile, the command line, or the IDE.
- A variable can be constrained at its declaration (`hour`).
- The domain of a decision variable can be tightened by replacing its type by a set of values of that type:
  - `x` must take an integer value.
  - `hour` must take an integer value between `0` and `23`.
  - `Taken` must be a subset of `{2,3,5,7,11,13}`.

Array and Set Comprehensions

An array or set can be built by a comprehension, using the notation $[\sigma \mid \gamma]$ or $\{\sigma \mid \gamma\}$, where σ is an expression evaluated for each element generated by the generator γ: a generator introduces one or more identifiers with values drawn from integer sets, optionally followed by a test.

Example

```
1 [x * 2 | x in 1..8]
2     evaluates to [2,4,6,8,10,12,14,16]
3 [x * y | x, y in 1..3 where x < y]
4     evaluates to [2,3,6]
5 [x + 2*y | x in 1..3, y in 1..4]
6     evaluates to [3,5,7,9,4,6,8,10,5,7,9,11]
7 {x + 2*y | x in 1..3, y in 1..4}
8     evaluates to {3,4,5,6,7,8,9,10,11}
```

Indexing: Syntactic Sugar

For example,

```
sum(i,j in 1..5)(i*j)
```

is syntactic sugar for

```
sum([i*j | i, j in 1..5])
```

This works for any function or predicate that takes an array as unique argument. In particular:

```
forall(i in 1..9)(x[i+1] = x[i] + y[i]);
```

is syntactic sugar for

```
forall([x[i+1] = x[i] + y[i] | i in 1..9]);
```

where `forall(array[int] of var bool: B)` is a function that returns the conjunction of all expressions in `B`: it generalises the 2-ary logical-and connective (`/\`).

Array Manipulation

- Changing the number of dimensions and their ranges, provided the numbers of elements match:

```
array1d(5..10,[|3,2|5,4|6,1|])
```

```
array2d(1..2,1..3,[2,7,3,7,4,9])
```

and so on, until `array6d`.

Tip: Try and keep your ranges starting from `1`:

  - It is easier to read a model under this usual convention.
  - Subtle errors may occur otherwise.

- Concatenation: for example, `[1,2] ++ [3,4]`.

Subtyping

A parameter can be used wherever a variable is expected. This extends to arrays: for example, a predicate or function expecting an argument of type `array[int] of var int` can be passed an argument of type `array[int] of int`.

The types `bool` and `int` are disjoint, but one can coerce from `bool` to `int` using the `bool2int` function: `bool2int(false) = 0` and `bool2int(true) = 1`.
This coercion is automatic but should be explicit for clarity.

Option Variables

An option variable is a decision variable that can also take the special value $\perp$, to be read "bottom".

A variable is declared optional with the keyword `opt`.

For example, `var opt 1..4: x` declares a variable `x` of domain `{1,2,3,4,⊥}`.

We will not cover the use of option variables in this course. However, one can see them:

- In the documentation:
  `var int` is a subtype of `var opt int`.
- In error messages:
  This is probably a sign that a model is too complicated.

Constraints

A constraint is the keyword `constraint` followed by a Boolean expression that must be true in every solution.

Example

```
1 constraint x < y;
2 constraint sum(Q) = 0 /\ alldifferent(Q);
3 constraint if x < y then x = y else x > y endif;
```

Constraints separated by a semi-colon (`;`) are implicitly connected by the 2-ary logical-and connective (`/\`).

What does `constraint x = x + 1` mean?
MiniZinc is declarative and has no destructive assignment: this equality constraint is not satisfied by any value for `x`.

Objective

The `solve` item gives the objective of the problem:

- `solve satisfy;`
  The objective is to solve a satisfaction problem.
- `solve minimize x;`
  The objective is to minimise the value of variable `x`.
- `solve maximize abs(x) * y;`
  The objective is to maximise the value of the expression `abs(x) * y`.

MiniZinc does not support multi-objective optimisation yet: multiple objective functions must either be aggregated into a weighted sum, or be handled outside a MiniZinc model.

Output Command

The `output` item prescribes what to print upon finding a solution: the keyword `output` is followed by a string array.

```
output [show(X)];
```

The function `show` returns a string representing the value of its argument expression.

```
output ["solution:"] ++ [if X[i]>0 then show(2*X[i])
    ++ ", " else " , " endif | i in 1..10];
```

The operator `++` concatenates two strings or two arrays.

`"X = \(X),"` is equivalent to `"X = "++show(X)++","`.

Tests

Conditional expressions can be formulated as follows:

- Conditional: `if` θ `then` ϕ_1 `else` ϕ_2 `endif`
- Generator: `[x | x in` ρ `where` θ `]`

The expressions ϕ_1 and ϕ_2 must have the same type.

The Boolean expression θ can depend on variables: with great power comes great responsibility!

Operators and Functions

- Booleans: `not`, `/\`, `\/`, `<->`, `->`, `<-`, `xor`, `forall`, `exists`, `xorall`, `iffall`, `clause`, `bool2int`
- Integers: `+`, `-`, `*`, `div`, `mod`, `abs`, `pow`, `min`, `max`, `sum`, `product`, `=`, `<`, `<=`, `=>`, `>`, `!=`, `..`
- Sets: `union`, `intersect`, `diff`, `symdiff`, `card`, `in`, `subset`, `superset`, `set2array`, `array_union`, `array_intersect`
- Strings: `++`, `concat`, `join`
- Arrays: `length`, `index_set`, `index_set_1of2`, `index_set_2of2`, `...`, `index_set_6of6`, `array1d`, `array2d`, `...`, `array6d`

Predicates and Functions

MiniZinc offers a large collection of predefined predicates and functions to enable a medium level at which models can be formulated: see Topic 3: Constraint Predicates.

Each predefined constrained function is defined by the use of the corresponding constraint predicate, possibly upon introducing a new variable.

Example

`count(A,v)>m` is replaced by `count(A,v,c) /\ c>m`.

It is also possible for modellers to define their own functions and predicates, as discussed on the next slide.

Predicate and Function Definitions

Example

```
1 function int: double(int: x);
2 function var int: double(var int: x);
3
4 predicate pos(var int: x);
5 function var bool: neg(var int: x);
```

A predicate can be used as a function returning `var bool`. For example, `bool2int(pos(a))` can be used.

Function and predicate names can be overloaded.

The body of a predicate or function definition is an expression of the same type as the returned value.

Example

```
1 function int: double(int: x) = 2 * x;
2 function var int: double(var int: x) = 2*x;
3
4 predicate pos(var int: x) = x > 0;
5 function var bool: neg(var int: x) = x < 0;
```

One can use `if ... then ... else ... endif`, predicates and functions, such as `forall` and `exists`, as well as `let` expressions (see next slide) in the body of a predicate or function definition.

Let Expressions

One can introduce local identifiers with a let expression.

Example

```
1  function int: double(int: x) =
2    let { int: y = 2 * x } in y;
3
4  function var int: double(var int: x) =
5    let { var int: y = 2 * x } in y;
6
7  function var int: double(var int: x) =
8    let { var int: y;
9          constraint y = 2 * x
10   } in y;
```

The 2nd and 3rd functions are equivalent; each use adds a decision variable to the model.

Constraints in Let Expressions

What is the difference between the next two definitions?

```
1 predicate posProd(var int: x, var int: y) =
2   let { var int: z; constraint z = x * y
3   } in z > 0;
4
5 predicate posProd(var int: x, var int: y) =
6   let { var int: z
7   } in z = x * y /\ z > 0;
```

Their behaviour is different in a negative context, such as `not posProd(a,b)`:

- The 1st one then ensures `a * b = z /\ z <= 0`.
- The 2nd one then ensures `a * b != z \/ z <= 0` and leaves `a` and `b` unconstrained.

Using Predicates and Functions

Advantages of using predicates and functions in a model:

- Software engineering good practice:
  - Reusability
  - Readability
  - Modularity
- The model might be solved more efficiently:
  - Better common-subexpression elimination.
  - The definitions can be technology- or solver-specific. If a predefined constraint predicate is a built-in of a solver, then its solver-specific definition is empty!

Remarks

- The order of model items does not matter.
- One can include other files.
  Example: `include "globals.mzn"`.
- The following functions are useful for debugging:
  - `assert`$(\theta,$`"error message"`$,\ \phi)$
    If the Boolean expression θ evaluates to `false`,
    then abort with the error message, otherwise return ϕ.
  - `trace(`"message"`,` $\phi$$)$
    Print the message and return ϕ.

UPPSALA UNIVERSITET

Outline

From a Problem to a Model

What is a good model for a constraint problem?

- A model that correctly represents the problem
- A model that is easy to understand and maintain
- A model that is solved efficiently, that is:
  - short solving time to find one, all, or best solution(s)
  - good solution within a limited amount of time
  - small search space (under constructive search)

Food for thought: What is correct, easy, short, good, . . . ?

Modelling Issues

Modelling is still more an Art than a Science:

- Choice of the decision variables
- Choice of the constraint predicates, in order to model the objective function, if any, and the constraints
- Optional for CP and LCG:
  - Choice of the consistency for each constraint
  - Choice of the variable selection strategy for search
  - Choice of the value selection strategy for search

  See Topic 8: Inference & Search in CP & LCG.

Make the model correct before making it efficient!

Choice of the Decision Variables

Example (Alphametic Problems)

SEND + MORE = MONEY:
Model without carry variables: 19 of 23 CP nodes visited:

$$1000 \cdot (S + M) + 100 \cdot (E + O) + 10 \cdot (N + R) + (D + E) = 10000 \cdot M + 1000 \cdot O + 100 \cdot N + 10 \cdot E + Y$$

Model with carry variables: 23 of 29 CP nodes are visited:

$$D + E = 10 \cdot C_1 + Y \ \wedge \ N + R + C_1 = 10 \cdot C_2 + E$$
$$\wedge \ E + O + C_2 = 10 \cdot C_3 + N \ \wedge \ S + M + C_3 = 10 \cdot M + O$$

GERALD + DONALD = ROBERT:
The model with carry variables is more effective in CP: only 791 of 869 nodes are visited, rather than 13,795 of 16,651 search nodes for the model without carry variables.

Choice of the Constraint Predicates

Example (The `alldifferent` constraint predicate)

The constraint `alldifferent([A[i] | i in 1..n])` usually leads to faster solving than its definition by a conjunction of $\Theta(n^2)$ disequality constraints:

```
forall(i,j in 1..n where i < j)(A[i]!=A[j])
```

For more examples, see Topic 3: Constraint Predicates.

Guidelines: Reveal Problem Structure

- Use few decision variables.
- Give tight domains to the decision variables.
- Avoid division constraints (`div` and `mod`).
- Avoid the disjunction of constraints (`\/`, `<-`, `->`, `<->`).
- Express the problem concisely
  (Topic 3: Constraint Predicates).
- Precompute solutions to a sub-problem into a table
  (Topic 3: Constraint Predicates, Topic 4: Modelling).
- Use implied constraints (Topic 4: Modelling).
- Use different viewpoints (Topic 4: Modelling).
- Exploit symmetries (Topic 7: Symmetry).

Careful: These guidelines of course have their exceptions!
It is important to test empirically several combinations of model, solver, and solving technology.

Use Few Decision Variables

When appropriate, use a single integer variable instead of an array of Boolean variables:

Example

Assume Joe must be assigned to exactly one task in `1..n`:

- Use a single integer variable, `var 1..n: joesTask`, representing *which* task Joe is assigned to.
- Don't use `array[1..n] of var bool:joesTask`, each element `joesTask[t]` representing *whether* (`true`) or not (`false`) Joe is assigned to task `t`, and `sum(t in 1..n)(bool2int(joesTask[t]))=1`.

When appropriate, use a single set variable instead of an array of Boolean or integer variables: see slides 46 and 48.

Give Tight Domains to the Variables

Without doing all the work of the solver, manually tightening the domains of the variables may accelerate the solving.
In particular, try and avoid `var int`:

Example

If the integer variable `s` represents the starting time of some task, then declare its domain with `var 0..horizon: s`, where `horizon` is suitably large, rather than `var int: s`.

Domain information may be used during flattening, so avoid setting a domain by constraints:

Counterexample

Do not reformulate `var 0..horizon: s` as follows:

```
var int: s; constraint 0<=s /\ s<=horizon;
```

Avoid Division Constraints

The `div` and `mod` functions often make the solving slow. Try and express division constraints otherwise, for instance by using a multiplication constraint or the `table` constraint predicate (see Topic 3: Constraint Predicates).

Example

```
constraint q = x div k;
constraint r = x mod k;
```

is logically equivalent to

```
constraint x = q * k + r;
var 0..k-1: r;  % better than 0 <= r < k
```

Avoid the Disjunction of Constraints

The disjunctive combination of constraints (with `\/`, `<-`, `->`, or `<->`) often makes the solving slow. Try and express disjunctive combinations of constraints otherwise.

Example

```
constraint x = 0 \/ x = 9;
```

is logically equivalent to

```
constraint x in {0,9};
```

and, even better, to

```
var {0,9}: x;
```

Example

```
constraint b -> x = 9;
constraint (not b) -> x = 0;
```

is logically equivalent to (recall that `bool2int(true)=1`)

```
constraint x = 9 * bool2int(b);
```

and to (note that array indexing starts at 1)

```
constraint x = [0,9][1+bool2int(b)];
```

But beware of such premature fine-tuning of a model!

The following versions are clearer and often good enough:

```
constraint x = if b then 9 else 0 endif;
```

and

```
constraint if b then x=9 else x=0 endif;
```

Express the Problem Concisely

Whenever possible, use a single predefined constraint predicate instead of a long-winded formulation.

Example (The `alldifferent` constraint predicate)

The constraint `alldifferent([A[i] | i in 1..n])` usually leads to faster solving than its definition by a conjunction of $\Theta(n^2)$ disequality constraints:

```
forall(i,j in 1..n where i < j)(A[i]!=A[j])
```

For more examples, see Topic 3: Constraint Predicates.

UPPSALA UNIVERSITET

Outline

Reification

Reification enables the reasoning about the truth of a constraint or a Boolean expression.

Example

```
constraint x < y;
```

requires that `x` be smaller than `y`.

```
constraint b <-> x < y;
```

requires that the Boolean variable `b` be `true` if and only if (iff) `x` is smaller than `y`: the constraint `x < y` is said to be reified, and `b` is called its reification.

Reification is a powerful mechanism that enables:

- higher-level modelling;
- easier implementation of the logical connectives.

The expression `bool2int(`γ`)`, for a constraint or Boolean expression γ, is an integer expression, with the truth of γ represented by `1` and its falsity by `0`.

Example (Cardinality constraint)

Constrain one or two of three constraints γ_1, γ_2, γ_3 to hold:
`1<=bool2int(`γ_1`)+bool2int(`γ_2`)+bool2int(`γ_3`)<=2.`

Reification comes with some drawbacks:

- Inference and relaxation may be poor.
- Not all constraints can be reified in MiniZinc.

Example (Soft Constraints: Photo Problem)

Consider `n` persons to be aligned in one row for a photo.

```
array[1..q,1..2] of 1..n: Pref;
```

Preference `k` in `1..q` states that person `Pref[k,1]` wants to be next to person `Pref[k,2]`.
Maximise the number of satisfied preferences.

Let decision variable `Pos[p]` denote the position in `1..n`, in left-to-right order, of person `p` in `1..n` on the photo.

The array `Pos` must form a permutation of the positions:

```
constraint alldifferent(Pos);
```

The objective is:

```
solve maximize sum(k in 1..q)
(          bool2int(abs(Pos[Pref[k,1]]-Pos[Pref[k,2]])=1));
```

Example (Soft Constraints: Photo Problem)

Consider `n` persons to be aligned in one row for a photo.

```
array[1..q,1..2] of 1..n: Pref;
```

Preference `k` in `1..q` states that person `Pref[k,1]` wants to be next to person `Pref[k,2]`.
Maximise the number of satisfied preferences.

Let decision variable `Pos[p]` denote the position in `1..n`, in left-to-right order, of person `p` in `1..n` on the photo.

The array `Pos` must form a permutation of the positions:

```
constraint alldifferent(Pos);
```

The objective is:

```
solve maximize sum(k in 1..q)
(          bool2int(abs(Pos[Pref[k,1]]-Pos[Pref[k,2]])=1));
```

Example (Soft Constraints: Photo Problem)

Consider `n` persons to be aligned in one row for a photo.

```
array[1..q,1..2] of 1..n: Pref;
```

Preference `k` in `1..q` states that person `Pref[k,1]` wants to be next to person `Pref[k,2]`.
Maximise the number of satisfied preferences.

Let decision variable `Pos[p]` denote the position in `1..n`, in left-to-right order, of person `p` in `1..n` on the photo.

The array `Pos` must form a permutation of the positions:

```
constraint alldifferent(Pos);
```

The objective is:

```
solve maximize sum(k in 1..q)
(          bool2int(abs(Pos[Pref[k,1]]-Pos[Pref[k,2]])=1));
```

Example (Soft Constraints: Photo Problem)

Consider `n` persons to be aligned in one row for a photo.

```
array[1..q,1..2] of 1..n: Pref;
```

Preference `k` in `1..q` states that person `Pref[k,1]` wants to be next to person `Pref[k,2]`.
Maximise the number of satisfied preferences.

Let decision variable `Pos[p]` denote the position in `1..n`, in left-to-right order, of person `p` in `1..n` on the photo.

The array `Pos` must form a permutation of the positions:

```
constraint alldifferent(Pos);
```

The objective is:

```
solve maximize sum(k in 1..q)
(          bool2int(abs(Pos[Pref[k,1]]-Pos[Pref[k,2]])=1));
```

Example (Soft Constraints: Weighted Photo Problem)

Consider `n` persons to be aligned in one row for a photo.

```
array[1..q,1..2] of 1..n: Pref;
array[1..q] of int: Weight
```

Preference `k` in `1..q` states that person `Pref[k,1]` wants to pay `Weight[k]` to be next to person `Pref[k,2]`.

Maximise the weighted number of satisfied preferences.

Let decision variable `Pos[p]` denote the position in `1..n`, in left-to-right order, of person `p` in `1..n` on the photo.

The array `Pos` must form a permutation of the positions:

```
constraint alldifferent(Pos);
```

The objective is:

```
solve maximize sum(k in 1..q)
(          bool2int(abs(Pos[Pref[k,1]]-Pos[Pref[k,2]])=1));
```

Example (Soft Constraints: Weighted Photo Problem)

Consider `n` persons to be aligned in one row for a photo.

```
array[1..q,1..2] of 1..n: Pref;
array[1..q] of int: Weight
```

Preference `k` in `1..q` states that person `Pref[k,1]` wants to pay `Weight[k]` to be next to person `Pref[k,2]`.

Maximise the weighted number of satisfied preferences.

Let decision variable `Pos[p]` denote the position in `1..n`, in left-to-right order, of person `p` in `1..n` on the photo.

The array `Pos` must form a permutation of the positions:

```
constraint alldifferent(Pos);
```

The objective is:

```
solve maximize sum(k in 1..q)
(Weight[k]*bool2int(abs(Pos[Pref[k,1]]-Pos[Pref[k,2]])=1));
```

Example (Sum of reified constraints)

The expression `sum(i in 1..n)(bool2int(A[i]=v))` denotes the number of elements of array `A` that equal `v`.

This idiom is very common in constraint-based models. So:

Example (The `count` constraint predicate)

The constraint `count(A,v,c)` holds iff variable `c` has the number of variables of array `A` that equal variable `v`.

For other predicates, see Topic 3: Constraint Predicates.

Example (The `count` constrained function)

The expression `count(A,v)` denotes the number of variables of array `A` that equal variable `v`.

Functional constraint predicates are available as functions.

UPPSALA UNIVERSITET

Outline

Motivating Example 1

Example (Agricultural experiment design, AED)

	plot1	plot2	plot3	plot4	plot5	plot6	plot7
barley							
corn							
millet							
oats							
rye							
spelt							
wheat							

Constraints to be **satisfied**:

1. Equal sample size: Every grain is grown in 3 plots.
2. Equal growth load: Every plot grows 3 grains.
3. Balance: Every grain pair is grown in 1 common plot.

Instance data: 7 rows, 7 cols, 3 grains, 3 plots, balance 1.

Motivating Example 1

Example (Agricultural experiment design, AED)

	plot1	plot2	plot3	plot4	plot5	plot6	plot7
barley	✓	✓	✓	–	–	–	–
corn	✓	–	–	✓	✓	–	–
millet	✓	–	–	–	–	✓	✓
oats	–	✓	–	✓	–	✓	–
rye	–	✓	–	–	✓	–	✓
spelt	–	–	✓	✓	–	–	✓
wheat	–	–	✓	–	✓	✓	–

Constraints to be **satisfied**:

1. Equal sample size: Every grain is grown in 3 plots.
2. Equal growth load: Every plot grows 3 grains.
3. Balance: Every grain pair is grown in 1 common plot.

Instance data: 7 rows, 7 cols, 3 grains, 3 plots, balance 1.

Example (BIBD *integer* model: ✓ $\rightsquigarrow$ 1 and – $\rightsquigarrow$ 0)

```
1  int: nbrVarieties; int: nbrBlocks;
2  set of int: Varieties = 1..nbrVarieties;
3  set of int: Blocks = 1..nbrBlocks;
4  int: sampleSize; int: blockSize; int: balance;
5  array[Varieties,Blocks] of var 0..1: BIBD;
6  solve satisfy;
7  constraint forall(v in Varieties)
8   (sampleSize = sum(b in Blocks)(BIBD[v,b]));
9  constraint forall(b in Blocks)
10  (blockSize = sum(v in Varieties)(BIBD[v,b]));
11 constraint forall(v, w in Varieties where v < w)
12  (balance = sum(b in Blocks)(BIBD[v,b]*BIBD[w,b]));
```

Example (Instance data for our AED)

```
1 nbrVarieties = 7; nbrBlocks = 7;
2 sampleSize = 3; blockSize = 3; balance = 1;
```

Example (Idea for another BIBD model)

barley	{plot1, plot2, plot3 }
corn	{plot1, plot4, plot5 }
millet	{plot1, plot6, plot7}
oats	{ plot2, plot4, plot6 }
rye	{ plot2, plot5, plot7}
spelt	{ plot3, plot4, plot7}
wheat	{ plot3, plot5, plot6 }

Constraints to be **satisfied**:

1. Equal sample size: Every grain is grown in 3 plots.
2. Equal growth load: Every plot grows 3 grains.
3. Balance: Every grain pair is grown in 1 common plot.

Example (BIBD *set* model: a block *set* per variety)

```
1  ...
2  ...
3  ...
4  ...
5  array[Varieties] of var set of Blocks: BIBD;
6  ...
7  ...
8   (sampleSize = card(BIBD[v]));
9  ...
10  (blockSize =
11         sum(v in Varieties)(bool2int(v in BIBD[b])));
12 ...
13  (balance = card(BIBD[v] inter BIBD[w]));
```

Example (Instance data for our AED)

```
1  ...
2  ...
```

Motivating Example 2[1]

Example (Hamming code: problem)

Toward high robustness in data transmission, we want to generate a codeword of m bits for each of the n symbols of an alphabet, such that the Hamming distance between any two codewords is at least some given constant d.

The Hamming distance between two same-length strings is the number of positions at which the two strings differ. Examples: $h(10001, 01001) = 2$ and $h(11010, 11110) = 1$.

ASCII has codewords of $m = 8$ bits for $n = 256$ symbols, but the least Hamming distance is $d = 1$: no robustness!

[1] Based on material by Christian Schulte

Example (Hamming code: model)

We encode a codeword of m bits as the set of positions of its unit bits, the least significant bit being at position 1. Example: 10001 is encoded as $\{1, 5\}$, and 01001 as $\{1, 4\}$. In general: $b_m \cdots b_1$ is encoded as $\{1 \cdot b_1, \ldots, m \cdot b_m\} \setminus \{0\}$. So the Hamming distance between two codewords is $u - i$, where u is the size of the union of their encodings and i is the size of the intersection of their encodings, that is the size of the symmetric difference of their encodings. Hence:

```
array[1..n] of var set of 1..m: C;
constraint forall(i, j in 1..n where i < j)
  (card(C[i] symdiff C[j]) >= d);
```

Definition

A set (decision) variable takes a set as value, and has a set of sets as domain. For its domain to be finite, a set variable must be a subset of a finite universe.

Set-constraint predicates exist for the following semantics:

- Cardinality: $|S| = n$
- Membership: $n \in S$
- Equality: $S_1 = S_2$
- Disequality $S_1 \neq S_2$
- Subset: $S_1 \subseteq S_2$
- Union: $S_1 \cup S_2 = S_3$
- Intersection: $S_1 \cap S_2 = S_3$
- Difference: $S_1 \setminus S_2 = S_3$
- Symmetric difference: $(S_1 \cup S_2) \setminus (S_1 \cap S_2) = S_3$
- Order: $S_1 \subseteq S_2 \vee \min((S_1 \setminus S_2) \cup (S_2 \setminus S_1)) \in S_1$
- Strict order: $S_1 \subset S_2 \vee \min((S_1 \setminus S_2) \cup (S_2 \setminus S_1)) \in S_1$

where the S_i are set variables and n is an integer variable.

Flatten with `-Gnosets` for a solver without set variables.